Práctica Quinta
RESOLUCION DE SISTEMAS DE ECUACIONES LINEALES

1.-Resolución utilizando RowReduce

**1.1.-El sistema se puede poner de la forma matricial A.x = b siendo A la matriz de los coeficientes ,
b el vector de los términos independientes y x el vector de las incógnitas**

In[1]:=

```
A = {{3, 2, 1}, {1, -1, 2}, {-1, 1, -1}};
b = {1, 2, 3};
x = {x1, x2, x3};
```

In[4]:=

```
A.x
```

Out[4]=

```
{3 x1 + 2 x2 + x3, x1 - x2 + 2 x3, -x1 + x2 - x3}
```

In[5]:=

```
Solve[A.x == b, x]
```

Out[5]=

```
{{x1 → -4, x2 → 4, x3 → 5}}
```

1.2.-También se podría haber puesto A.x = b y multiplicando por A^{-1} tendríamos que x = A^{-1}.b

In[6]:=

```
x = Inverse[A].b
```

Out[6]=

```
{-4, 4, 5}
```

En ambos casos, la matriz A es regular es decir

In[7]:=

```
Det[A]
```

Out[7]=

```
-5
```

In[8]:=

```
RowReduce[A]
```

Out[8]=

```
{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}
```

Con lo que el rango es 3, y por tanto tiene Inversa

1.3.-Apliquemos la reducción por filas a la matriz ampliada, que es la matriz A a la que añadimos una columna más que es la de los términos independientes

In[9]:=

```
traspuesta = Transpose[A];
aumentar = AppendTo[traspuesta, b];
MatrixForm[RowReduce[Transpose[aumentar]]]
```

Out[11]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & -4 \\ 0 & 1 & 0 & 4 \\ 0 & 0 & 1 & 5 \end{pmatrix}$$

Observamos que la última columna es la de las soluciones, puesto que utiliza el métod de Gauss_Jordan, de eliminación

1.4.-Supongamos que el número de ecuaciones es superior al de incógnitas

In[12]:=

```
B = {{3, 2, 1}, {1, -1, 2}, {-1, 1, -1}, {4, 1, 3}, {2, 3, 0}};
c = {1, 2, 3, 3, 4};
traspuesta = Transpose[B];
aumentar = AppendTo[traspuesta, c];
MatrixForm[RowReduce[Transpose[aumentar]]]
```

Out[16]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & -4 \\ 0 & 1 & 0 & 4 \\ 0 & 0 & 1 & 5 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

Observamos que hay dos filas combinación lineal y que el resultado es el mismo que el ejemplo anterior

1.5.-Supongamos el siguiente sistema

In[17]:=

```
B = {{3, 2, 1}, {1, -1, 2}, {4, 1, 3}, {2, 3, 0}};
c = {1, 2, 1, 4};
```

Si lo resolvemos tendremos

In[19]:=

```
traspuesta = Transpose[B];
aumentar = AppendTo[traspuesta, c];
MatrixForm[RowReduce[Transpose[aumentar]]]
```

Out[21]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

La cuarta fila nos dice que los rangos de la matriz de los coeficientes y de la ampliada son distintos y por tanto el sistema es incompatible.

2.-Problema

Dado un sistema de ecuaciones, en función de parámetros m y n, calcularlos para que el sistema sea compatible y determinado (por ejemplo)

$x - y + 2 = 0$

$x + 3y - 1 = 0$

$(m + 2n)x + (2m - 1 - n)y + 5m + 1 = 0$

$mx - (m + n + 4)y = 5$

In[22]:=

```
Clear[x, y, m, m];
ec1 = x - y + 2 == 0;
ec2 = 2 x + 3 y - 1 == 0;
ec3 = (m + 2 n) x + (2 m - 1 - n) y + 5 m + 1 == 0;
ec4 = m x -  (m + n + 4) y == 5;
```

Veamos si las rectas ec1 y ec2 se cortan

In[27]:=

```
solucion = Solve[{ec1, ec2}, {x, y}]
```

Out[27]=

```
{{x → -1, y → 1}}
```

Entonces qué condiciones tienen que darse en ec3 y ec4 para que pasen por el punto (-1, 1)

In[28]:=

```
condicion1 = ec3 /. solucion[[1]];
condicion2 = ec4 /. solucion[[1]];
```

In[30]:=

```
sistema = {condicion1, condicion2}
```

Out[30]=

```
{6 m - 3 n == 0, -4 - 2 m - n == 5}
```

In[31]:=

```
Solve[sistema, {m, n}]
```

Out[31]=

$$\left\{\left\{m \to -\frac{9}{4},\ n \to -\frac{9}{2}\right\}\right\}$$

3.-Problema

Discutir el sistema para los distintos valores de λ y de μ

In[32]:=

```
Clear[sistema];
B = {{1, 2, 1}, {1, 1, 2}, {1, 3, λ}, {1, 2, 1}};
c = {2, 3, 1, μ};
```

In[35]:=

```
traspuesta = Transpose[B]
```

Out[35]=

{{1, 1, 1, 1}, {2, 1, 3, 2}, {1, 2, λ, 1}}

In[36]:=

```
aumentar = AppendTo[traspuesta, c]
```

Out[36]=

{{1, 1, 1, 1}, {2, 1, 3, 2}, {1, 2, λ, 1}, {2, 3, 1, μ}}

In[37]:=

```
sistema = Transpose[aumentar]
```

Out[37]=

{{1, 2, 1, 2}, {1, 1, 2, 3}, {1, 3, λ, 1}, {1, 2, 1, μ}}

In[38]:=

```
MatrixForm[sistema]
```

Out[38]//MatrixForm=

$$\begin{pmatrix} 1 & 2 & 1 & 2 \\ 1 & 1 & 2 & 3 \\ 1 & 3 & \lambda & 1 \\ 1 & 2 & 1 & \mu \end{pmatrix}$$

In[39]:=

```
Det[sistema]
```

Out[39]=

$2\lambda - \lambda\mu$

In[40]:=

```
Solve[Det[sistema] == 0, {λ, μ}]
```

Solve::svars : Equations may not give solutions for all "solve" variables. More…

Out[40]=

$\{\{\lambda \to 0\}, \{\mu \to 2\}\}$

Casos :

a) $\mu = 2$, $\lambda = 0$

b) $\mu = 2$, $\lambda \neq 0$

c) $\mu \neq 2$, $\lambda = 0$

In[41]:=

```
uno = sistema /. {μ → 2, λ → 0}
```

Out[41]=

```
{{1, 2, 1, 2}, {1, 1, 2, 3}, {1, 3, 0, 1}, {1, 2, 1, 2}}
```

In[42]:=

```
RowReduce[uno]
```

Out[42]=

```
{{1, 0, 3, 4}, {0, 1, -1, -1}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

In[44]:=

```
dos = sistema /. μ → 2
```

Out[44]=

```
{{1, 2, 1, 2}, {1, 1, 2, 3}, {1, 3, λ, 1}, {1, 2, 1, 2}}
```

In[46]:=

```
RowReduce[dos]
```

Out[46]=

```
{{1, 0, 0, 4}, {0, 1, 0, -1}, {0, 0, 1, 0}, {0, 0, 0, 0}}
```

In[47]:=

```
tres = sistema /. λ → 0
```

Out[47]=

```
{{1, 2, 1, 2}, {1, 1, 2, 3}, {1, 3, 0, 1}, {1, 2, 1, μ}}
```

In[48]:=

```
RowReduce[tres]
```

Out[48]=

```
{{1, 0, 3, 0}, {0, 1, -1, 0}, {0, 0, 0, 1}, {0, 0, 0, 0}}
```
